
Supplementary Data 1
SAS/IML SINGLE-sample rarefaction program.

Written by M. Kowalewski.

_____________________________________________________________________________________________

************************************************************************

DESCRIPTION

This code performs a single-sample rarefaction or sub-sampling depending on the definition of macro-variables listed directly below. A dataset should consist of a single column of values. Each row represents a different taxon (e.g., species). Each value represents a number of specimens representing that species. The number of rows with non-zero values represents sample diversity (taxon richness).

************************************************************************

SAS/IML Code written by Michal Kowalewski (Virginia Tech). This version last updated on February 17, 2005

************************************************************************

NOTES

1. This code performs a single-sample rarefaction or sub-sampling depending on the definition of macro-variables listed directly below.

2. The dataset should consist of a single column of values.

3. Each row represents a different taxon (e.g., species).

4. Each value represents number of specimens representing that species.

5. The number of rows with non-zero values represents sample diversity (taxon richness).

6. Zero values are allowed, but will be disregarded.

7. The sum of values across rows represents the total number of specimens (sample size).

WARNINGS

1. At small sub-sampling levels, species richness (sub-sampled diversity) may be 1.

Consequently, H and Hmax are both zero and E cannot be computed (and is reported as missing value).

Concurrently, D and De will be 1.

2. The outputs are not rounded.

3. For larger samples (>1000 specimens), running time-efficient simulation can be achieved either by setting low iteration number (100 is usually sufficient to get decent estimates) or by using a fast desktop.

4. Start with a small number of iterations (times=10) to make sure the program works and to assess how long it takes to run iterations. Note that the computational time increases at a non-linear rate with the number of iterations. That is, if it takes 1 minute to run 100 iterations, 1000 iterations typically will take MUCH MORE time than 10 minutes.

EXPLANATION OF SYMBOLS USED IN THE OUTPUTS

OUTPUT 1: The raw re-sampling output showing results for each iteration (WORK.REPORT)

diver - species richness for a given subsample in a given iteration

maxtax - the percentage abundance of the most abundant taxon

H - Shannon Diversity Index

E - Shannon Evenness Index

D - Simpson Diversity Index

De - Simpson Evenness Index

n - subsampled sample size

iter - iteration number

OUTPUT 2: The summary estimates from the simulations (WORK.FINAL2)

n - subsampled sample size

iter - number of iterations

maxtax - the percentage abundance of the most abundant taxon

div - mean diversity (averaged across a given set of iterations)

H - Shannon Diversity Index

E - Shannon Evenness Index

D - Simpson Diversity Index

De - Simpson Evenness Index

std_div - standard deviation of a given variable (e.g., std of diversity)

P_div2_5 - lower 95% confidence limit (2.5 percentile) for a given variable

P_div25 - lower 50% confidence limit (25 percentile) for a given variable

P_div75 - upper 50% confidence limit (75 percentile) for a given variable

P_div97_5 - upper 50% confidence limit (97.5 percentile) for a given variable

SE_P_div - mean diversity plus 1 standard error

SE_M_div - mean diversity minus 1 standard error

********************************************************************;

* THE CODE BEGINS HERE;

OPTIONS linesize=256 pagesize=5000;

*--ENTER THESE MACROVARIABLES;

%let times=1000; * - number of iterations;

%let raref='yes'; * - type "'yes'" to generate a rarefaction curve and 'no' to perform subsampling standardization;

*NOTE!If "choose='no'" you must specify the macrovariable "number" defined below (any integer smaller than your sample size);

*--RAREFACTION MACROVARIABLES;

%let cut=6; * - rarefaction sample step (an integer);

%let auto='yes'; * - type "'yes'" to carry out a rarefaction up to the actual sample size;

*NOTE! If "auto='no'" then define macrovariable 'max' below);

%let max=100; * - maximum rarefaction sample size;

*--SUBSAMPLING MACROVARIABLES;

%let number=30; * - a standardized sample size (any pre-selected integer value SMALLER THAN your original sample);

**********************************************************************;

*ENTER DATA

Note 1: Replace a column of grey-highlighted numbers with your data.

Note 2: You can simply copy-and-paste from Excel or other file containing your data.

Note 3: No missing values are allowed.

Note 4: Zero values are allowed.

Note 5: Only integers are allowed.

After completing all above steps you should be ready to run your data. Click on a running man on the SAS taskbar;

**********************************************************************;

DATA new; * this data step will generate a sas-dataset called 'new';

        infile cards;

        input loc1;

cards;

0

0

2

0

0

1

0

2

0

13

84

0

;

run;

Title1'Single-sample rarefaction (SAS/IML)';

Title2'written by M. Kowalewski, 02/17/2005';

Title3'diversity indices of the actual sample';

PROC iml; * - start iml code;

USE new;

READ all var _num_ into V;

X=t(V);

START weight (X,Y); * - MODULES FOR ITERATIVE RESAMPLING;

  k=ncol(X);

   do i=1 to k;

     s=X[1,i];

     z=repeat(i,1,s);

     zz=zz||z;

   end;

   zzz=t(zz);

   Y=zzz;

FINISH weight;

START ranvec(in,frac,v_out);

      k=nrow(in);

      v_index=in;

   do i=1 to k;

      rand=floor((k-i+1)*ranuni(0) + 1);

      v_ran=v_ran||v_index[rand];

      v_index=remove(v_index,rand);

   end;

      temp=t(v_ran[,1:frac]);

      v_out=temp;

FINISH ranvec;

START recr(X,frac,out);

      k=ncol(X);

      run weight(X,Y);

      run ranvec(Y,frac,YY);

      do i=1 to k;

      a=i#(YY=i);

          ab=sum(a)/i;

      abun=abun//ab;

      end;

      out=abun;

FINISH recr;

START INDEX(Y,S,H); * - MODULES USED FOR COMPUTING EVENNESS METRICS;

  X=Y;

  S=nrow(X);

  P=X/sum(X);

  H=-sum(P#log(P));

FINISH INDEX;

START FINAL(X,out);

        Y=X;

                Z=(Y/shape(sum(Y),nrow(Y),1));

                D=1/sum(Z#Z);

                De=D/nrow(Y);

        Ymax=shape(sum(Y)/nrow(Y),nrow(Y),1);

        run index(Y,S,H);

        run index(Ymax,Smax,Hmax);

        if Hmax=0 then E=.;

                else E=H/Hmax;

        out=H||E||D||De;

FINISH FINAL;

START STATS(X,outstat); * - MODULES EXECUTING THE RAREFACTION SIMULATION;

    Z=t(X); * - data (counts of specimens);

    Div=ncol(loc(Z>0)); * - number of taxa in a sample;

        N=sum(Z); * - number of specimens in a sample;

        truenonzero=Z[loc(Z>0)];

        truemax=max(truenonzero)/sum(truenonzero);

        run final(truenonzero,even);

        t1=even[,1];

        t2=even[,2];

        t3=even[,3];

        t4=even[,4];

                print 'actual sample diversity' Div;

        print 'actual sample size' N;

                print 'shannon diversity index H' t1;

                print 'shannon evenness index E' t2;

                print 'Simpson diversity index D' t3;

                print 'Simpson evenness index De' t4;

                print 'the most abundant taxon (%)' truemax;

    do i=1 to &times;

         if &auto='yes' then max=N;

         else max=&max;

         do j=&cut to max by &cut;

        run recr(X,j,random);

                DivR=ncol(loc(random>0));

                Max=max(random)/sum(random);

                nonzero=random[loc(random>0)];

                run final(nonzero,indices);

                outt=DivR||Max||indices||j||i;

        output=output//outt;

         end;

    end;

       outstat=output;

FINISH STATS;

START STAT2(X,outstat); * - MODULES EXECUTING THE SUBSAMPLING SIMULATION;

    Z=t(X); * - data (counts of specimens);

    Div=ncol(loc(Z>0)); * - number of taxa in a sample;

        N=sum(Z); * - number of specimens in a sample;

        truenonzero=Z[loc(Z>0)];

        run final(truenonzero,even);

        t1=even[,1];

        t2=even[,2];

        t3=even[,3];

        t4=even[,4];

        truemax=max(truenonzero)/sum(truenonzero);

                print 'actual sample diversity' Div;

        print 'actual sample size' N;

                print 'shannon diversity index H' t1;

                print 'shannon evenness index E' t2;

                print 'Simpson diversity index D' t3;

                print 'Simpson evenness index De' t4;

                print 'the most abundant taxon (%)' truemax;

        do i=1 to &times;

     do j=&number to &number;

        run recr(X,j,random);

                DivR=ncol(loc(random>0));

                Max=max(random)/sum(random);

                nonzero=random[loc(random>0)];

                run final(nonzero,indices);

                outt=DivR||Max||indices||j||i;

        output=output//outt;

      end;

    end;

       outstat=output;

FINISH STAT2;

if &raref='yes' then do;

RUN STATS(X,outstat);

end;

else do;

RUN stat2(X,outstat);

end;

create random from outstat;

append from outstat;

close random;

QUIT;

run;

data report;

        set random;

                diver=col1;

                                maxtax=col2;

                                H=col3;

                                E=col4;

                                D=col5;

                                De=col6;

                n=col7;

                iter=col8;

                drop col1-col8;

proc sort;

        by n;

Title3'OUTPUT 1: Raw resampling output sorted by n (number of specimens) and i (number of iterations)';

proc print;

proc univariate noprint;

        var diver maxtax H E D De;

        by n;

        output out=final1 n=iter mean=div maxtax H E D De std=std_div std_max std_H std_E std_D std_De

                          PCTLPRE=P_div P_max P_H P_E P_D P_De PCTLPTS=2.5 25 75 97.5;

data final2;

set final1;

SE_P_div=div+std_div;

SE_M_div=div-std_div;

Title3'OUTPUT 2: Summary estimates of species richness and diversity indices';

proc print;

proc plot;

plot div*n='.';

Title3'rerafaction curve (in case of subsampling, only one data point will show)';

run;

quit;

_____________________________________________________________________________________________

SAS/IML MULTI-sample rarefaction program.

Written by M. Kowalewski.

************************************************************************

DESCRIPTION

This code performs a multi-sample rarefaction. A dataset should consist of columns of values. Each column represents a sample and each row represents a different taxon (e.g., species). Each value represents a number of specimens representing that species. The number of rows with non-zero values represents sample diversity (taxon richness).

************************************************************************

SAS/IML Code written by Michal Kowalewski (Virginia Tech). This version last updated on February 17, 2005
************************************************************************

NOTES

1. This code performs a multi-sample rarefaction.

2. The dataset should consist of columns of values.

3. Each row represents a different taxon (e.g., species).

4. Each value represents number of specimens representing that species.

5. The number of rows with non-zero values represents sample diversity (taxon richness).

6. Zero values are allowed, but will be disregarded.

7. The sum of values across rows represents the total number of specimens (sample size).

WARNINGS

1. The outputs are not rounded.

2. For larger samples (>1000 specimens), running time-efficient simulation can be achieved either by setting low iteration number (100 is usually sufficient to get decent estimates) or by using a fast desktop.

3. Start with a small number of iterations (times=10) to make sure the program works and to assess how long it takes to run iterations. Note that the computational time increases at a non-linear rate with the number of iterations. That is, if it takes 1 minute to run 100 iterations, 1000 iterations typically will take MUCH MORE time than 10 minutes.

EXPLANATION OF SYMBOLS USED IN THE OUTPUTS

OUTPUT 1: The summary estimates from the simulations:

n - subsampled sample size

iter - number of iterations

div - mean diversity (averaged across a given set of iterations)

std_div - standard deviation of a given variable (e.g., std of diversity)

P_div2_5 - lower 95% confidence limit (2.5 percentile) for a given variable

P_div25 - lower 50% confidence limit (25 percentile) for a given variable

P_div75 - upper 50% confidence limit (75 percentile) for a given variable

P_div97_5 - upper 50% confidence limit (97.5 percentile) for a given variable

********************************************************************;

* THE CODE BEGINS HERE;

OPTIONS linesize=256 pagesize=5000;

*--ENTER THESE MACROVARIABLES;

%let times=100;  * - number of iterations to estimate confidence intervals;
*--RAREFACTION MACROVARIABLES;

%let sam=3; * - number of samples to be included in the analysis;
%let cut=3; * - standardized number of specimens per sample;
**********************************************************************;

*ENTER DATA

Note 1: Replace columns of grey-highlighted numbers with your data.

Note 2: You can simply copy-and-paste from Excel or other file containing your data.

Note 3: Modify in according to number of columns present in your dataset, the Loc1-Loc4 line (e.g., if your dataset is composed by 9 columns you should write: loc.1-loc.9).

Note 4: No missing values are allowed.

Note 5: Zero values are allowed.

Note 6: Only integers are allowed.

After completing all above steps you should be ready to run your data. Click on a running man on the SAS taskbar;

**********************************************************************;

Title1'Multi-sample rarefaction (SAS/IML)';
Title2'written by M. Kowalewski, 12/02/2004';
DATA new;
        infile cards;
        input loc1-loc4;
cards; * this data step will generate a sas-dataset called 'new';
1 3 17 5
2 4 0 4
5 5 0 143
6 6 0 0
17 7 45 0
42 4 8 12
6 2 32 0
0 9 194 0
32 0 4 0
0 17 1 1
1 0 0 2
;
run;
PROC iml;
USE new;
READ all var _num_ into V;
START weight (X,Y);
  k=ncol(X);
   do i=1 to k;
     s=X[1,i];
     z=repeat(i,1,s);
     zz=zz||z;
   end;
   zzz=t(zz);
   Y=zzz;
FINISH weight;
START ranvec(in,frac,v_out);
      k=nrow(in);
      v_index=in;
   do i=1 to k;
      rand=floor((k-i+1)*ranuni(0) + 1);
      v_ran=v_ran||v_index[rand];
      v_index=remove(v_index,rand);
   end;
      temp=t(v_ran[,1:frac]);
      v_out=temp;
FINISH ranvec;
START recr(X,frac,out);
      Z=t(X);
      k=ncol(Z);
      run weight(Z,Y);
      run ranvec(Y,frac,YY);
      do i=1 to k;
      a=i#(YY=i);
          ab=sum(a)/i;
      abun=abun//ab;
      end;
          *out=abun[loc(abun>0),];
      out=abun;
FINISH recr;
START FINAL(X,out);
 tot=&sam*&cut; *-total number of specimens that will be resampled;
 Z=X; * - original data matrix (columns=samples) (rows=species);
 index=t(1:ncol(Z));
 do i=1 to &times;
  run ranvec(index,&sam,sam);
  rand=Z[,sam]; *-subset of samples, with number of samples defined by the macrovariable "sam";
  do j=1 to ncol(rand);
   a=rand[,j];
   run recr(a,&cut,random);
   b=b||random;
   c=b[,+];
 end;
   b=shape(0,nrow(rand),1);
   DivR=ncol(loc(c>0));
   outt=DivR||&sam||&cut||tot||i;
   output=output//outt;
end;
 out=output;
FINISH FINAL;
RUN FINAL(V,out);
create random from out;
append from out;
close random;
QUIT;
run;
data report;
        set random;
                div=col1;
                samples=col2;
                sam_n=col3;
                tot_n=col4;
                iter=col5;
                drop col1-col5;
proc print;
proc univariate noprint;
        var div;
        output out=final1 n=iter mean=diver std=std PCTLPRE=P_ PCTLPTS=2.5 25 75 97.5;
proc print;
run;
quit;

